LAPACK  3.10.0 LAPACK: Linear Algebra PACKage

## ◆ cpptri()

 subroutine cpptri ( character UPLO, integer N, complex, dimension( * ) AP, integer INFO )

CPPTRI

Purpose:
``` CPPTRI computes the inverse of a complex Hermitian positive definite
matrix A using the Cholesky factorization A = U**H*U or A = L*L**H
computed by CPPTRF.```
Parameters
 [in] UPLO ``` UPLO is CHARACTER*1 = 'U': Upper triangular factor is stored in AP; = 'L': Lower triangular factor is stored in AP.``` [in] N ``` N is INTEGER The order of the matrix A. N >= 0.``` [in,out] AP ``` AP is COMPLEX array, dimension (N*(N+1)/2) On entry, the triangular factor U or L from the Cholesky factorization A = U**H*U or A = L*L**H, packed columnwise as a linear array. The j-th column of U or L is stored in the array AP as follows: if UPLO = 'U', AP(i + (j-1)*j/2) = U(i,j) for 1<=i<=j; if UPLO = 'L', AP(i + (j-1)*(2n-j)/2) = L(i,j) for j<=i<=n. On exit, the upper or lower triangle of the (Hermitian) inverse of A, overwriting the input factor U or L.``` [out] INFO ``` INFO is INTEGER = 0: successful exit < 0: if INFO = -i, the i-th argument had an illegal value > 0: if INFO = i, the (i,i) element of the factor U or L is zero, and the inverse could not be computed.```

Definition at line 92 of file cpptri.f.

93 *
94 * -- LAPACK computational routine --
95 * -- LAPACK is a software package provided by Univ. of Tennessee, --
96 * -- Univ. of California Berkeley, Univ. of Colorado Denver and NAG Ltd..--
97 *
98 * .. Scalar Arguments ..
99  CHARACTER UPLO
100  INTEGER INFO, N
101 * ..
102 * .. Array Arguments ..
103  COMPLEX AP( * )
104 * ..
105 *
106 * =====================================================================
107 *
108 * .. Parameters ..
109  REAL ONE
110  parameter( one = 1.0e+0 )
111 * ..
112 * .. Local Scalars ..
113  LOGICAL UPPER
114  INTEGER J, JC, JJ, JJN
115  REAL AJJ
116 * ..
117 * .. External Functions ..
118  LOGICAL LSAME
119  COMPLEX CDOTC
120  EXTERNAL lsame, cdotc
121 * ..
122 * .. External Subroutines ..
123  EXTERNAL chpr, csscal, ctpmv, ctptri, xerbla
124 * ..
125 * .. Intrinsic Functions ..
126  INTRINSIC real
127 * ..
128 * .. Executable Statements ..
129 *
130 * Test the input parameters.
131 *
132  info = 0
133  upper = lsame( uplo, 'U' )
134  IF( .NOT.upper .AND. .NOT.lsame( uplo, 'L' ) ) THEN
135  info = -1
136  ELSE IF( n.LT.0 ) THEN
137  info = -2
138  END IF
139  IF( info.NE.0 ) THEN
140  CALL xerbla( 'CPPTRI', -info )
141  RETURN
142  END IF
143 *
144 * Quick return if possible
145 *
146  IF( n.EQ.0 )
147  \$ RETURN
148 *
149 * Invert the triangular Cholesky factor U or L.
150 *
151  CALL ctptri( uplo, 'Non-unit', n, ap, info )
152  IF( info.GT.0 )
153  \$ RETURN
154  IF( upper ) THEN
155 *
156 * Compute the product inv(U) * inv(U)**H.
157 *
158  jj = 0
159  DO 10 j = 1, n
160  jc = jj + 1
161  jj = jj + j
162  IF( j.GT.1 )
163  \$ CALL chpr( 'Upper', j-1, one, ap( jc ), 1, ap )
164  ajj = real( ap( jj ) )
165  CALL csscal( j, ajj, ap( jc ), 1 )
166  10 CONTINUE
167 *
168  ELSE
169 *
170 * Compute the product inv(L)**H * inv(L).
171 *
172  jj = 1
173  DO 20 j = 1, n
174  jjn = jj + n - j + 1
175  ap( jj ) = real( cdotc( n-j+1, ap( jj ), 1, ap( jj ), 1 ) )
176  IF( j.LT.n )
177  \$ CALL ctpmv( 'Lower', 'Conjugate transpose', 'Non-unit',
178  \$ n-j, ap( jjn ), ap( jj+1 ), 1 )
179  jj = jjn
180  20 CONTINUE
181  END IF
182 *
183  RETURN
184 *
185 * End of CPPTRI
186 *
subroutine xerbla(SRNAME, INFO)
XERBLA
Definition: xerbla.f:60
logical function lsame(CA, CB)
LSAME
Definition: lsame.f:53
complex function cdotc(N, CX, INCX, CY, INCY)
CDOTC
Definition: cdotc.f:83
subroutine csscal(N, SA, CX, INCX)
CSSCAL
Definition: csscal.f:78
subroutine chpr(UPLO, N, ALPHA, X, INCX, AP)
CHPR
Definition: chpr.f:130
subroutine ctpmv(UPLO, TRANS, DIAG, N, AP, X, INCX)
CTPMV
Definition: ctpmv.f:142
subroutine ctptri(UPLO, DIAG, N, AP, INFO)
CTPTRI
Definition: ctptri.f:117
Here is the call graph for this function:
Here is the caller graph for this function:
{"ft_lang_label":"__label__en","ft_lang_prob":0.7709686,"math_prob":0.92334974,"size":1274,"snap":"2021-43-2021-49","text_gpt3_token_len":419,"char_repetition_ratio":0.1,"word_repetition_ratio":0.05172414,"special_character_ratio":0.3265306,"punctuation_ratio":0.1358885,"nsfw_num_words":0,"has_unicode_error":false,"math_prob_llama3":0.97588867,"pos_list":[0,1,2],"im_url_duplicate_count":[null,null,null],"WARC_HEADER":"{\"WARC-Type\":\"response\",\"WARC-Date\":\"2021-10-24T21:03:31Z\",\"WARC-Record-ID\":\"<urn:uuid:1a360a3c-dea4-4ecd-8320-f19a189e227f>\",\"Content-Length\":\"27990\",\"Content-Type\":\"application/http; msgtype=response\",\"WARC-Warcinfo-ID\":\"<urn:uuid:a3d3246c-5c0c-4a7e-a594-6156b902d1af>\",\"WARC-Concurrent-To\":\"<urn:uuid:74ce342c-923c-4d60-a834-676d52556082>\",\"WARC-IP-Address\":\"160.36.131.221\",\"WARC-Target-URI\":\"https://www.netlib.org/lapack/explore-html/d3/db9/group__complex_o_t_h_e_rcomputational_gac3b364c7ab658b9716b14cf05e735d9d.html\",\"WARC-Payload-Digest\":\"sha1:WVDPHMMEC5WC2RNV5ARDJ6FIUGPMILFS\",\"WARC-Block-Digest\":\"sha1:454WH6CUKYMOXVQSL5QXHFCTUCWOHYO5\",\"WARC-Identified-Payload-Type\":\"application/xhtml+xml\",\"warc_filename\":\"/cc_download/warc_2021/CC-MAIN-2021-43/CC-MAIN-2021-43_segments_1634323587606.8_warc_CC-MAIN-20211024204628-20211024234628-00459.warc.gz\"}"}